

DOCUMENT RESUME

ED 258 896

SO 016 648

AUTHOR Burns, Edward
 TITLE Loan and Mortgage Spreadsheet.
 PUB DATE 85
 NOTE 25p.
 PUB TYPE Reports - Descriptive (141) -- Computer Programs (101)

EDRS PRICE MF01/PC01 Plus Postage.
 DESCRIPTORS Computer Software; *Consumer Education; Decision Making; *Economics Education; Higher Education; *Interest (Finance); *Loan Repayment; *Money Management; Secondary Education

ABSTRACT

Information about the effect of changes in interest rates and the length of repayment on mortgages is available through the use of this loan and mortgage spreadsheet program designed for use on Apple computers. It is especially designed to help people evaluate the relative advantages of a variable rate mortgage with a low initial rate of interest against mortgages with a higher but fixed rate of interest. The program allows the user to determine monthly payments for both types of mortgages and to calculate total interest payments. The program has two major options, one for fixed rate mortgages and one for variable rate mortgages. For fixed rate mortgages up to ten different interest rates can be specified. For adjustable rate mortgages the program allows the user to experiment with different rates of increase and decrease. Procedures for operating the program are detailed and the program is provided. (IS)

 * Reproductions supplied by EDRS are the best that can be made *
 * from the original document. *

U.S. DEPARTMENT OF EDUCATION
NATIONAL INSTITUTE OF EDUCATION
EDUCATIONAL RESOURCES INFORMATION
CENTER (ERIC)

- This document has been reproduced as received from the person or organization originating it.
- Minor changes have been made to improve reproduction quality.
- Points of view or opinions stated in this document do not necessarily represent official NIE position or policy.

"PERMISSION TO REPRODUCE THIS
MATERIAL HAS BEEN GRANTED BY

Edward Burns

TO THE EDUCATIONAL RESOURCES
INFORMATION CENTER (ERIC)."

ED 258896

LOAN AND MORTGAGE SPREADSHEET

by

Edward Burns

State University of New York at Binghamton

1985

SD 016 648

There was a time in days gone-by when deciding how to finance the purchase of a home was relatively simple. Interest rates were not only fairly uniform between different lending institutions, but the home buyer did not face the prospect of paying astronomical monthly payments and/or payments that fluctuated from one year to the next.

The home buyer of today is faced with a maze of options that can make the financing of a home a bewildering experience. As a result of several recent "creative" trends in home financing, a home buyer might be faced with interest rates that range between 10 to 15 percent. The following shows how different interest rates can affect monthly payments on a \$100,000 mortgage for a term of 30 years:

<u>INTEREST</u>	<u>MONTHLY PAYEMENT</u>
10%	\$877.57
11%	\$952.32
12%	\$1028.61
13%	\$1106.2
14%	\$1184.87
15%	\$1264.44

The problem is not finding the lowest possible interest rate to finance a home; the problem involves deciding whether obtaining a loan with an initial low interest rate is worth the risk that the rate might increase in future years...and in some instances the increase could be drastic.

A major decision that a home buyer is faced with is whether to obtain a fixed-rate mortgage or adjustable rate mortgage (ARM). An ARM offers the home buyer an interest rate at what is usually an initial low level. At specified periods that interest rate is adjusted based on some index over which the lending institution has no control (e.g., Treasury securities index).

Consider this example of an ARM for a \$100,000 30 year loan. The initial rate might be 11.5% which would result in monthly payments of \$990.29. If this loan were a three year ARM, the bank or savings and loan would have the option of adjusting the interest rate after three years of payments have been made. Assume that this loan also has a 2% "cap" which means that the interest rate could not be increased by more than 2% for each adjustment period (and not all ARM's have caps!). After three years the new adjusted interest rate is 13.5% and the monthly payment is now \$1140.02. After three more years the interest rate is adjusted again to 15.5% and the monthly payment increased to \$1288.7.

The dilemma faced by the home buyer is very clear. Should the home buyer obtain a fixed-rate mortgage, say at 14%, that would result in fixed monthly payments of \$1184.87? Or should the home buyer secure an ARM as described above that would have initial monthly payments of \$990.29, but that could increase to \$1288.7 after six years?

Using the Program

The Loan and Mortgage Spreadsheet program will help the home buyer take some of the guesswork out of home financing. The program allows you to determine monthly payments for both fixed-rate and adjustable rate loans. For each type of loan, the amount of the loan (i.e., principal), the term of the loan, and/or the interest rates can be changed to experiment with different loan situations.

The program has two major options. The first option allows for the experimentation with fixed-rate mortgages. The following sample run illustrates program output for a \$75,000 fixed-rate loan for 30 years. For each fixed-rate mortgage up to 10 different interest rates can be specified. The fixed-rate option is initiated by entering F after the prompt

FIXED OR ADJUST. INTEREST (F OR A)?

The amount of the loan or the initial principal is then entered. In this example the loan is for \$75,000. The term of the loan is initially set at 30 years. The different interest rates to be considered are then entered. After the last interest rate is entered, the RETURN key is pressed and the different program options for modifying variables will appear at the bottom of the screen

Figure 1 about here



4

In Figure 1 seven different fixed-rate interest amounts are shown, ranging from 14.125% to 14.755%. Many newspapers print a weekly summary of both fixed-rate and adjustable rate mortgages. For a \$75,000 30 year loan, the monthly payments will range from \$896.08 to 933.66. For each interest rate specified, the total interest paid over the life of the loan is also printed. For the first interest rate of 14.125%, where the monthly payment is \$896.08, the total interest paid over the life of the loan is \$247,588.6. The total interest is the number of payments made (or number of years indicated multiplied by 12) times the monthly payment value minus the initial principal.

To see what will happen if the term of the loan is 20 rather than 30 years, Y is entered after the prompt at the bottom of the screen and then 20 entered after the NUMBER OF YEARS prompt. The monthly payments and total interest values are then recalculated and printed for each of the fixed-rates specified. The output for a \$75,000 20 year loan for the rates indicated are shown in Figure 2.

Figure 2 about here

The effects of changing the loan amount can be tested by entering L after the variable decision prompt and then entering the new loan amount. As shown in Figure 3, the loan amount has been reduced from \$75,000 to \$50,000 for

a 30 year term. As soon as the loan amount and term of the loan are changed, the monthly payments and total interest for each of the fixed-rates is recalculated and printed.

7 Figure 3 about here

If desired new interest rates can be entered using the loan amount and term previously specified. This is accomplished by entering I after the prompt at the bottom of the screen.

The adjustable rate mortgage option is initiated by entering A after the initial program prompt. If you are already in the program, entering M after the variable decision prompt will send control to the beginning of the program and the initial program prompt.

The method for determining monthly payments for the adjustable rate option is as follows: after an adjustment has been made (an increase or decrease in the interest rate) monthly payments are calculated using the remaining balance for the remaining term for the new interest rate specified. Take note that there is no industry standard for calculating adjustable rate mortgages, and how interest rates are adjusted and/or what index is used to make the adjustment will vary from institution to institution.

To illustrate a typical adjustable mortgage, consider

a 3 year adjustable \$75,000 30 year loan in which the initial interest rate is 12.75%. The interest rate of 1 entered for the first three years to indicate the interest rate for those years. To determine what would happen if the interest rate is increased to 14.75% in the fourth year, the 4th interest rate entered is 14.75. As shown in Figure 4, the monthly payment increases from \$815.02 to \$929.02 as a result of the two percent increase in the interest rate.

 Figure 4 about here

The scenario shown in Figure 4 represents a situation in which there is a two percent adjustment cap and a 16 percent loan cap. This means that no loan adjustment can be greater than a two percent increase, and that the maximum interest rate for the loan cannot exceed 16 percent. The output in Figure 4 illustrates the worst possible scenario or what would happen if all the maximum increases allowable took place. The last interest rate of 16% indicates that the interest rate for the remaining 24 years of the loan is 16 percent.

Because different interest rates are used, the total amount of interest paid over the term of the loan must be



determined for each of the different interest rates used and then added. The total interest rate for the years specified in the loan is printed at the top of the screen.

As with fixed-rate mortgages, the program allows for considerable experimentation when dealing with adjustable mortgages. Figure 5 shows what occurs when the initial loan is reduced from \$75,000 to \$50,000 for the adjustable rates already specified.

Figure 5 about here

The examples so far shown really reflect quite favorably on adjustable rate mortgages. But this is where the plot thickens. There are many different ways of constructing adjustable mortgage rates, and one dandy way from a lender's viewpoint is not to have any caps. Assume that you have an ARM with an initial interest rate of 12 percent in the first year that increases to 20 percent in the ninth year. What happens to your monthly payments in this situation? By entering the interest rate as shown in Figure 6 you can see that the monthly payment increases from \$771.46 a month in the first year to \$1236.05 in the ninth year.

Figure 6 about here

The example in Figure 7 is even a little more drastic. Here the initial interest rate is 12 percent for the first three years, increases to 17.5 percent for the second three years, and then to 23 percent for the remainder of the loan. As can be seen, the monthly payment for the seventh and remaining years is \$1,416.76.

 Figure 7 about here

Rather than considering gloomy scenarios, data can be entered that reflect less than maximum increases or even decreases in interest rates. What is important is that the program allows one to experiment with different possibilities with respect to future ARM interest rates. There is no answer as to what type of mortgage is preferable. However, your Apple and this program should reduce some of the mystery surrounding adjustable rate mortgages.

How the Program Works

The type of loan to be considered is entered in line 120. An entry of A indicates an adjustable mortgage and F signifies a fixed-rate mortgage. The amount of the loan is entered in line 220 and the value is stored in the variable AL. The term of the loan is entered in line 250 and the value stored in NY.

Up to 10 different interest rates can be entered. Each interest rate is entered via line 370, first stored in YI\$, and then stored in the YI(C) array, where C can be a value from 1 to 10.

If fixed-interest rates are being considered, and F is equal to F , the monthly payment for each of the interest rates specified is calculated in line 630:

$$630. PA = AL * (MI / (1 - (1 + MI)^{-NP}))$$

The variables in line 630 are as follows: PA = monthly payment, AL = the amount of the loan, MI = the interest for a monthly period (which is calculated in line 490), and NP = number of payments (which is calculated in line 270).

The total amount paid for a fixed-rate mortgage is determined in line 790:

$$790 T = INT (NP * PA * 100 + .5) / 100$$

The total amount is found by multiplying the number of payments (NP) by the amount of each payment (PA). The total interest paid is determined in the print statement in line 790 by subtracting the initial amount of the loan from the total amount paid or $T - AL$.

If the adjustable mortgage option is selected, monthly payments are calculated in line 630 until an adjustment in interest rates is made; that is, there is a difference between

interest rates for two given years. This is determined in line 520:

520 IF YI(C) = YI(C - 1) THEN 630

If the interest rates between any two years is different, the interest paid prior to that change is calculated in lines 580 and 590. Line 580 is used to calculate the balance of the loan still owed. The amount of interest paid for a given period having the same interest rate is then determined in line 590.

For adjustable loans, the number of payments to be made is determined in line 600. The variable CP signifies the number of years for a loan subperiod or the number of continuous years having the same interest rate. As an example, assume that in the fourth year the interest rate for an adjustable loan changes. Monthly payments are then calculated for the remaining principal or money owed after three years for a term of 27 years.

The output for adjustable loans is printed in line 800. Before the total interest is printed in line 1010, the interest for the remaining years of the loan is calculated in line 970. A variable decision is then entered in line 1050. If the value entered and stored in ND\$ is equal to M, control is transferred to line 60 and the program begins anew; if ND\$ is

equal to L, the loan amount (AL) variable in line 220 and the number of years variable (NY) in line 250 can be redefined; if ND\$ is equal to Y, only the number of years variable is redefined; if ND\$ is equal to I, control is shifted to line 270 and new interest rates are entered; finally, if ND\$ is equal to E, control is transferred to line 1210 and the program is terminated.

The major loop for calculating monthly payments and total interest paid for each of the interest rates specified begins in line 420 and extends to line 840:

```
420 FOR X = 1 TO NC
```

```
840 NEXT X
```

After the last interest rate to be considered has been entered and the RETURN key is pressed, NC is set to the number of variables entered (line 900). Each time loan information is redefined, 1 to NC monthly payments are calculated.

Entering the Program and Modifications


When entering the program it is probably a good idea not to enter the ONERR statement in line 50 until the program has been debugged. Should a syntax or some other type of error occur when entering the program, you will want to know exactly what the error is rather than have control shift to another part of the program.

As was said before, there is no standard for calculating adjustable rate mortgages. Rather than having a cap on the maximum adjustment interest increase, there might be a cap on the maximum dollar amount increase. Thus, it is possible in this situation that the monthly payment does not pay the full interest due thereby resulting in negative amortization (the principal increases rather than decreases). Another possibility is that the term of the loan is increased after an adjustment in the interest rate is made. Whatever the case might be, modify the Loan and Mortgage Spreadsheet program so that it functions in accordance with the method for calculating adjustable rate mortgages in your area (a tip: give a call to the chief loan officer at your local bank).

Finally, when using the Loan and Mortgage Spreadsheet program remember that other variables might be a factor in deciding the best type of mortgage for a given situation. As an example, a bank might charge "points" for the privilege of obtaining a loan. If the points for a loan were 3 and the loan was for \$50,000, then \$1,500 would be paid at the time the loan is secured. Obviously a mortgage with points might not be as good a buy as a mortgage from a different lender that did not charge points.

There does indeed seem an infinite number of ways that

a lender can separate you from your money. But armed with your Apple and Loan and Mortgage Spreadsheet program, you can minimize that separation as much as possible.



AMOUNT OF LOAN? 75000
NUMBER OF YEARS? 30

YEARLY INTEREST?

INTEREST	PAYMENTS	TOTAL INTEREST
14.125%	\$896.08	\$247588.6
14.24%	\$903.52	\$250265.52
14.375%	\$910.96	\$252946.05
14.5%	\$918.42	\$255630.1
14.575%	\$922.89	\$257242.19
14.755%	\$933.66	\$263702.57

M=MORTGAGE, L=LOAN, Y=YEARS, I=INTEREST
ENTER M, L, Y OR I (E TO EXIT):

Figure 1

AMOUNT OF LOAN? 75000
NUMBER OF YEARS? 20

INTEREST	PAYMENTS	TOTAL INTEREST
14.125%	\$939.46	\$150469.74
14.25%	\$946.29	\$152109.41
14.375%	\$953.14	\$153752.71
14.5%	\$960	\$155399.6
14.575%	\$964.12	\$156389.43
14.755%	\$974.04	\$158770.18
14.875%	\$980.67	\$160361.33

M=MORTGAGE, L=LOAN, Y=YEARS, I=INTEREST
ENTER M, L, Y OR I (E TO EXIT):

Figure 2

10
AMOUNT OF LOAN? 50000
NUMBER OF YEARS? 30

INTEREST	PAYMENTS	TOTAL INTEREST
14.125%	\$597.39	\$165059.06
14.25%	\$602.34	\$166843.68
14.375%	\$607.31	\$168630.7
14.5%	\$612.28	\$170420.07
14.575%	\$615.26	\$171494.79
14.755%	\$622.44	\$174077.43
14.875%	\$627.23	\$175801.71

M=MORTGAGE, L=LOAN, Y=YEARS, I=INTEREST
ENTER M, L, Y OR I (E TO EXIT):

Figure 3

AMOUNT OF LOAN? 75000
NUMBER OF YEARS? 30

TOTAL INTEREST = \$276071.64

INTEREST	YEAR	PAYMENTS
12.75%	1	\$815.02
12.75%	2	\$815.02
12.75%	3	\$815.02
14.75%	4	\$929.91
14.75%	5	\$929.91
14.75%	6	\$929.91
16%	7	\$1000.88

M=MORTGAGE, L=LOAN, Y=YEARS, I=INTEREST
ENTER M, L, Y OR I (E TO EXIT):

Figure 4

AMOUNT OF LOAN? 50000
NUMBER OF YEARS? 30

TOTAL INTEREST = \$184047.76

INTEREST	YEAR	PAYMENTS
12.75%	1	\$543.35
12.75%	2	\$543.33
12.75%	3	\$543.35
14.75%	4	\$619.94
14.75%	5	\$619.94
14.75%	6	\$619.94
16%	7	\$667.25

M=MORTGAGE, L=LOAN, Y=YEARS, I=INTEREST
ENTER M, L, Y OR I (E TO EXIT):

Figure 5

AMOUNT OF LOAN? 75000
NUMBER OF YEARS? 30

TOTAL INTEREST = \$344853

INTEREST	YEAR	PAYMENTS
12%	1	\$771.46
13%	2	\$829.05
14%	3	\$886.93
15%	4	\$944.99
16%	5	\$1003.18
17%	6	\$1061.44
18%	7	\$1119.7
19%	8	\$1177.92
20%	9	\$1236.05

M=MORTGAGE, L=LOAN, Y=YEARS, I=INTEREST
ENTER M, L, Y OR I (E TO EXIT):

Figure 6

AMOUNT OF LOAN? 75000
NUMBER OF YEARS? 30

TOTAL INTEREST = \$400050.76

INTEREST	YEAR	PAYMENTS
12%	1	\$771.46
12%	2	\$771.46
12%	3	\$771.46
17.5%	4	\$1090.28
17.5%	5	\$1090.28
17.5%	6	\$1090.28
23%	7	\$1416.76

M=MORTGAGE, L=LOAN, Y=YEARS, I=INTEREST
ENTER M, L, Y OR I (E TO EXIT):

Figure 7

21

```

10 REM LOAN AND MORTGAGE SPREADSHEET
20 REM
30 REM INPUT TYPE OF MORTGAGE
40 REM
50 ONERR GOTO 1190
60 HOME : CLEAR
70 DIM YI(10)
80 FOR K = 1 TO 39:SP$ = SP$ + " ": NEXT K
90 VTAB 8: FOR K = 1 TO 39: PRINT "-";: NEXT K
100 PRINT
110 VTAB 1
120 INPUT "FIXED OR ADJUST. INTEREST (F OR A)? ";T$
130 IF T$ = "A" THEN 160
140 IF T$ = "F" THEN 210
150 GOTO 100
160 VTAB 1: PRINT SP$
170 VTAB 1: HTAB 12: FLASH : PRINT "ADJUSTABLE MORTGAGE": NORMAL
180 REM
190 REM INPUT LOAN DATA
200 REM
210 VTAB 3
220 INPUT "AMOUNT OF LOAN? ";AL
230 LL = AL
240 VTAB 4: PRINT SP$: VTAB 4
250 INPUT "NUMBER OF YEARS? ";NY
260 VTAB 1: PRINT SP$
270 NP = NY * 12
280 IF ND$ = "Y" OR ND$ = "I" THEN AL = LL
290 C = 1:CP = 0
300 IF ND$ < > "I" THEN 320
310 ND = 0: FOR K = 12 TO 22: VTAB K: PRINT SP$: NEXT K
320 CP = 0: VTAB 6: PRINT SP$
330 VTAB 6
340 IF ND < 1 THEN 360
350 GOTO 420
360 YI$ = ""
370 INPUT "YEARLY INTEREST? ";YI$
380 CP = CP + 1
390 IF YI$ = "" THEN 900
400 YI(C) = VAL (YI$)
410 GOTO 490
420 FOR X = 1 TO NC
430 IF ND = 1 THEN CP = CP + 1
440 REM
450 REM MI = INTEREST PER MONTH
460 REM PA = MONTHLY PAYMENT
470 REM T = TOTAL INTEREST AND PRINCIPAL PAID
480 REM
490 MI = YI(C) / 100 / 12
500 IF T$ = "F" THEN 630

```

22

```

510 IF C = 1 THEN 630
520 IF YI(C) = YI(C - 1) THEN 630
530 REM
540 REM CALCULATE MONTHLY PAYMENT
550 REM
560 PM = (CP - 1) * 12
570 PI = YI(C - 1) / 100 / 12
580 BA = PA * (1 - (1 + PI) ^ (PM - NP)) / PI
590 TI = TI + (CP - 1) * 12 * PA - (AL - BA)
600 NP = NP - (CP - 1) * 12
610 CP = 1
620 AL = BA
630 PA = AL * (MI / (1 - (1 + MI) ^ - NP))
640 RP = INT (PA * 100 + .5) / 100
650 REM
660 REM PRINT INTEREST RATES
670 REM
680 PRINT
690 IF T$ = "F" THEN 760
700 VTAB 10
710 PRINT TAB( 7)"INTEREST YEAR PAYMENTS"
720 VTAB 11 + C: PRINT SP$
730 VTAB 11 + C
740 PRINT TAB( 8);YI(C)"%" TAB( 19);C; TAB( 25)"$"RP
750 GOTO 810
760 VTAB 10
770 PRINT "INTEREST PAYMENTS TOTAL INTEREST"
780 VTAB 11 + C: PRINT SP$: VTAB 11 + C
790 T = INT (NP * PA * 100 + .5) / 100
800 PRINT TAB( 2);YI(C)"%" TAB( 11)"$"RP; TAB( 23)"$"T - AL
810 C = C + 1
820 IF C > 10 THEN 900
830 IF ND < 1 THEN 880
840 NEXT X
850 C = C - 1:CP = 0
860 CP = 0
870 GOTO 940
880 VTAB 6: HTAB 17: PRINT SP$: VTAB 6
890 GOTO 330
900 ND = 1:C = C - 1:NC = C:CP = 0
910 REM
920 REM VARIABLE CONTROL DECISION
930 REM
940 FOR K = C TO 1 STEP - 1
950 IF YI(K) < > YI(K - 1) THEN 970
960 C = C - 1: NEXT K
970 TI = TI + (NY * 12 - (C - 1) * 12) * PA - AL
980 TI = INT (TI * 100 + .5) / 100
990 IF T$ = "F" THEN 1020
1000 VTAB 6: PRINT SP$

```



```

1010 VTAB 6: PRINT "TOTAL INTEREST = $"TI
1020 TI = 0
1030 VTAB 22
1040 PRINT "M=MORTGAGE, L=LOAN, Y=YEARS, I=INTEREST "
1050 VTAB 23: INPUT "ENTER M, L, Y OR I (E TO EXIT):";ND$
1060 VTAB 22: PRINT SP$: PRINT SP$
1070 IF ND$ = "M" THEN 60
1080 IF ND$ = "L" THEN 1130
1090 IF ND$ = "Y" THEN 240
1100 IF ND$ = "I" THEN 270
1110 IF ND$ = "E" THEN 1210
1120 GOTO 1030
1130 VTAB 1
1140 FOR K = 1 TO 5: PRINT SP$: NEXT K
1150 GOTO 210
1160 REM
1170 REM ERROR AND EXIT MESSAGES
1180 REM
1190 VTAB 22: INVERSE : PRINT "INPUT ERROR: PRESS 'RETURN' TO CONTINUE.":
NORMAL
1200 GET E$: GOTO 1030
1210 VTAB 22: PRINT "THIS PROGRAM IS NOW OVER."
1220 END

```

